
Read-me-file: Extending the GTAP Database and Model to 

Cover Domestic Support Issues using the EU as example 

Kirsten Urban, Hans Grinsted Jensen and Martina Brockmeier 

 

 

1. General remarks: 

This approach consists of various steps to update domestic support in the GTAP database. We 

start the procedure with a *.tab-file to adjust the aggregation of PSE data (compare Section 

2.). Figure 1 provides a first overview of the different steps, which are initiated thereafter. The 

allocation of PSE data is described in Section 3 and is followed by the calculation of shocks 

needed to update the allocation of domestic support in the GTAP database (compare Section 

4.). Thereafter, we start an iterative procedure to implement the allocation of domestic support 

in the GTAP database using the altertax program (compare Section 5.). A GTAP view file is 

used to produce view files for tax rates and some coefficients (compare Section 6.). After 

obtaining the updated final database, you first run a numeraire shock scenario to implement a 

new closure (compare Section 7.). Then, calculate new shocks and run the policy simulations, 

which in this case correspond to a 100% removal of the other transfer payments (see Section 

8.). In the following sections, we briefly describe the *.tab-files and identify all input, output 

and windows batch files, which are necessary to run the model.  

2. ds_regmap.tab 

This tab.file is used to adjust the PSE data at EU member state level to the aggregation chosen 

in GTAP. 

Input: 

 pse_eu.har   PSE data for 27 EU member states (DSDATA) 

 sets.har   GTAP sets file (GTAPSETS) 

 ds_sets.har   Additional sets for domestic support (DSSETS) 

 ds_regmap.tab  Tablo input file to change the aggregation of PSE data 

 ds_regmap.cmf  Gempack command file  

Output: 

 pse_eu_agg.har  Aggregated PSE data 

Windows batch file: 

 dsregmap.bat 

Figure 1: Overview update domestic support procedure 

 

Source: Authors’ elaboration. 

3. ds.tab 

This ds.tab-file is used to calculate the shares of PSE data in relation to the EUROSTAT 

production values to allow fir the scaling procedure within the GTAP database.  

Input: 

 pse_eu_agg.har  Aggregated PSE data (DSDATA) 

 sets.har   GTAP sets (GTAPSETS) 

 ds_sets.har   Additional sets for domestic support (DSSETS) 

 ds.tab    Tab.file to calculate shares 

 ds.cmf    Command.file 

Output: 

 ds_gtap.har   Allocation of PSE data 

Windows batch file: 

 ds.bat 

You can run the ds.tab file using the windows batch file "ds.bat" As result you will then 

obtain "ds_gtap.har". 

In the file ds.tab you have to decide between option (a) and (b) to allocate the Single Farm 

Payment (SFP): 

(a) Homogenous rate across factors and across sectors 

 
!------------------------------------------------------------------------------! 

! 1. option                                                                    ! 

!------------------------------------------------------------------------------! 

! Homogenous rate across factors and across sectors (default)                  ! 

!------------------------------------------------------------------------------! 

 

Coefficient (all,i,NNAT_COMM)(all,j,TRAD_COMM)(all,r,REG) TVFM(i,j,r)  

    #Factor Usage #; 

 

Formula (all,i,NNAT_COMM)(all,j,TRAD_COMM)(all,r,REG) 

    TVFM(i,j,r)=sum{k,NNATDIS_COMM:FACTMAP(k)=i,EVFA(k,j,r)}; 

 

Coefficient (all,i,NNAT_COMM)(all,r,REG) TVFMSHR(i,r)  

    # Factor Usage Shares #; 

 

Formula (all,i,NNAT_COMM)(all,r,REG) 

    TVFMSHR(i,r) = sum{j,AGRI_COMM,TVFM(i,j,r)}/sum{k,NNAT_COMM, 

                   sum{j,AGRI_COMM,TVFM(k,j,r)}}; 

 

Coefficient (all,i,NNAT_COMM)(all,r,DS_REG) OTRAN(i,r)  

    # Other Transfers in the form of Activity-generic Payments #; 

 

Formula (all,i,NNAT_COMM)(all,r,DS_REG) 

    OTRAN(i,r) = 0; 

 

Formula (all,i,NNAT_COMM)(all,r,DS_REG) 

    OTRAN(i,r) = TVFMSHR(i,r)*sum{k,OTPM,OTPH(k,r)}; 
 
(b) OTP is allocated with homogenous rate across primary agricultural commodities to the 

factors land, labor and capital according to predetermined shares based on assumptions. The 

assumptions can be adjusted by diluting the share of the factors labor and capital in equal 

proportions and loading this onto the factors land to increase the degree of decoupling. The 

percental reduction of the factors labor and capital can be modified using the coefficient 

SHIFTFCT in pse_eu.har. 

!------------------------------------------------------------------------------! 

! 2a. option                                                                   ! 

!------------------------------------------------------------------------------! 

! Reducing the factor share of capital and labor, shift this to onto the factor! 

! land                                                                         ! 

!------------------------------------------------------------------------------! 

 

Coefficient (all,i,NNAT_COMM)(all,j,TRAD_COMM)(all,r,REG) TVFM(i,j,r)  

    #Factor Usage #; 

 

Formula (all,i,NNAT_COMM)(all,j,TRAD_COMM)(all,r,REG) 

    TVFM(i,j,r)=sum{k,NNATDIS_COMM:FACTMAP(k)=i,EVFA(k,j,r)}; 

 

Coefficient (all,i,NNAT_COMM)(all,r,REG) TVFMSHR(i,r)  

    # Factor Usage Shares #; 

 

Formula (all,i,NNAT_COMM)(all,r,REG) 

    TVFMSHR(i,r) = sum{j,AGRI_COMM,TVFM(i,j,r)}/sum{k,NNAT_COMM, 

                   sum{j,AGRI_COMM,TVFM(k,j,r)}}; 

 

Coefficient(all,i,NNAT_COMM) SHIFTFCT(i) 

    # predetermined reduction of the shares for capital and labor #; 

    Read (all,i,NNAT_COMM)SHIFTFCT(i) from file DOMSUP header "SFCT"; 

 

Coefficient (all,i,MFAC_COMM)(all,r,DS_REG) RFCTSHR(i,r) 

    # Reduced factor share #; 

Formula (all,i,MFAC_COMM)(all,r,DS_REG) 

    RFCTSHR(i,r)=TVFMSHR(i,r)*SHIFTFCT(i); 

 

Coefficient (all,i,NNAT_COMM)(all,r,DS_REG) DECOUPSHR(i,r) 

    # Degree of Decoupling #; 

Formula (all,i,MFAC_COMM)(all,r,DS_REG) 

    DECOUPSHR(i,r)=RFCTSHR(i,r); 

Formula (all,r,DS_REG) 

    DECOUPSHR("land",r)= 1-sum{k,MFAC_COMM,RFCTSHR(k,r)}; 

 

Coefficient (all,i,NNAT_COMM)(all,r,DS_REG) OTRAN(i,r)  

    # Other Transfers in the form of Activity-generic Payments #; 

 

Formula (all,i,NNAT_COMM)(all,r,DS_REG) 

    OTRAN(i,r) = 0; 

 

Formula (all,i,NNAT_COMM)(all,r,DS_REG) 

    OTRAN(i,r) = DECOUPSHR(i,r)*sum{k,OTPM,OTPH(k,r)}; 

 

(c) Adjust the factor usage share manually in the har-file according to your assumption of the 

allocation of SFP or create a header in the file pse_eu_agg.har to read in the factor usage 

share. 
 

!------------------------------------------------------------------------------! 

! 2b. option                                                                   ! 

!------------------------------------------------------------------------------! 

 

 

coefficient (all,i,NNAT_COMM)(all,r,REG) TVFMSHR(i,r)  

    # Factor Usage Shares #; 

 

Formula (all,i,NNAT_COMM)(all,r,REG) 

    TVFMSHR(i,r) = 0; 

 

Formula (all,r,REG) 

    TVFMSHR("land",r) = 1; 

COEFFICIENT (all,i,NNAT_COMM)(all,r,DS_REG) OTRAN(i,r)  

    # Other Transfers in the form of Activity-generic Payments #; 

 

FORMULA (all,i,NNAT_COMM)(all,r,DS_REG) 

    OTRAN(i,r) = 0; 

 

FORMULA (all,i,NNAT_COMM)(all,r,DS_REG) 

    OTRAN(i,r) = TVFMSHR(i,r)*sum{k,OTPM,OTPH(k,r)}; 

 

!------------------------------------------------------------------------------! 

! Calculation of the OTP /SFP share of production value                        ! 

!------------------------------------------------------------------------------! 

 

Zerodivide (zero_by_zero) default 0.0; 

Zerodivide (nonzero_by_zero) default 0.0; 

 

Coefficient (all,i,NNATDIS_COMM)(all,r,DS_REG) OTP_PVS(i,r)  

    # OTP production value share #; 

 

Formula (all,r,DS_REG) 

    OTP_PVS("land",r)       = OTRAN("land",r)/sum{k,AGRI_COMM,PRODN(k,r)}; 

Formula (all,r,DS_REG) 

    OTP_PVS("capital",r)    = OTRAN("capital",r)/sum{k,AGRI_COMM,PRODN(k,r)}; 

Formula (all,r,DS_REG) 

    OTP_PVS("unsklab",r)    = OTRAN("labor",r)*SHARELAB("unsklab",r) 

                            / sum{k,AGRI_COMM,PRODN(k,r)}; 

Formula (all,r,DS_REG) 

    OTP_PVS("sklab",r)      = OTRAN("labor",r)*SHARELAB("sklab",r) 

                            / sum{k,AGRI_COMM,PRODN(k,r)}; 

4. ds_shk.tab 

This tab.file is used to calculate the shocks to adjust the GTAP database.  

Input: 

 basedata.har   GTAP Version 8 base data (GTAPDATA) 

 ds_gtap.har   Allocation of PSE data (DSDATA) 

 sets.har   GTAP sets file (GTAPSETS) 

 ds_sets.har   Additional sets for domestic support (DSSETS) 

 ds_shk.tab   Tab file to calculate the shocks 

 ds_shk.cmf   Command file 

Output: 

 ds_shk.har   Shock file 

Windows batch file:  

 dsshk.bat 

According to your chosen allocation of the SFP, you have to adjust the code in this tab.file: 

(a) Homogenous rate (Standard GTAP database) 

!------------------------------------------------------------------------------! 

! 1. option: Homogenous rate (Standard GTAP database)                         ! 

!------------------------------------------------------------------------------! 

 

Coefficient (all,r,DS_REG) OTP_SHR(r) 

    # Share of domestic support allocated to SFP subsidies #; 

Formula(initial) (all,r,DS_REG) 

    OTP_SHR(r) = sum{i,NNATDIS_COMM,sum{j,AGRI_COMM,VFMT(i,j,r) 

               - VFMX(i,j,r,"otp")}} 

               / sum{a,AGRI_COMM,VOM(a,r)}; 

 

Coefficient (all,i,NNATDIS_COMM)(all,r,DS_REG) OTP_PVS(i,r) 

    # OTP production value share #; 

    Read OTP_PVS from file DSDATA header "OPVS"; 

 

Coefficient (all,r,DS_REG) SHOTP(r) 

    # Shock to OTP (SFP) #; 

Formula (all,r,DS_REG) 

    SHOTP(r) = 0; 

 

Formula (all,r,DS_REG) 

    SHOTP(r) = sum{i,NNATDIS_COMM,OTP_PVS(i,r)} - OTP_SHR(r); 

 

(b) According to the assumed allocation across factors 

!------------------------------------------------------------------------------! 

! 2. option: According to assumed allocation across factors                    ! 

!------------------------------------------------------------------------------! 

 

COEFFICIENT (all,i,NNATDIS_COMM)(all,r,DS_REG) OTP_SHR(i,r) 

    # Share of domestic support allocated to SFP subsidies #; 

FORMULA(INITIAL) (all,i,NNATDIS_COMM)(all,r,DS_REG) 

    OTP_SHR(i,r) = sum{j,AGRI_COMM,VFMT(i,j,r) - VFMS(i,j,r)} 

                 / sum{a,AGRI_COMM,VOM(a,r)}; 

 

COEFFICIENT (all,i,NNATDIS_COMM)(all,r,DS_REG) OTP_PVS(i,r) 

    # OTP production value share #; 

    read OTP_PVS from file DSDATA header "OPVS"; 

COEFFICIENT (all,i,NNATDIS_COMM)(all,r,DS_REG) SHOTP(i,r) 

    # Shock to OTP (SFP) #; 

FORMULA (all,i,NNATDIS_COMM)(all,r,DS_REG) 

    SHOTP(i,r) = 0; 

 

FORMULA (all,i,NNATDIS_COMM)(all,r,DS_REG) 

    SHOTP(i,r) = OTP_PVS(i,r)- OTP_SHR(i,r); 

 

First, you need to run this file to calculate the new value flows VFMT, VFMX, VDFMX and 

VIFMX. Second, import the new value flows into new header in basedata.har. Then re-

activate the "read into" command in the tab-file and re-run ds_shk.tab. 

e.g.: 

Coefficient (all,i,ENDW_COMM)(all,j,PROD_COMM)(all,r,REG)(all,g,BASEGROUP) 

    VFMX(i,j,r,g) 

    # VFM + ACT and GCT #; 

FORMULA(initial) (all,i,ENDW_COMM)(all,j,PROD_COMM)(all,r,REG)(all,g,BASEGROUP)  

    VFMX(i,j,r,g) = VFM(i,j,r) + FTRV(i,j,r);! 

    Read VFMX from file GTAPDATA header "VFMX"; 

! 

5. ds_slug.tab 

Start the update procedure using a variant of the altertax program (Malcolm, 1998). In this 

model we use a variant of the altertax code where all endowments are sluggish (original tab-

file is called slug.tab). Using ds_slug.tab the GTAP database is updated stepwise with a new 

allocation of domestic support.  

Input: 

 basedata.har GTAP base data (GTAPDATA) 

 sets.har GTAP sets file (GTAPSETS) 

 ds_sets.har Additional sets for domestic support (DSSETS) 

 ds_shk1.har 

 ds_shk8.har Shock files 

 ds_slug.tab Altertax tab file 

 ds_slug.sti Stored Input file 

Output: 

 altbase.upd, altbase1.upd, …,  

altbase8.upd    Updated GTAP database 

Windows batch file: 

 update_ds.bat (for the whole procedure up to now) 

Again, you have to select one of the SFP allocation options:  

Please select either tfsfp(r) or tfsfp(i,r) according to your allocation  

of SFP 

Equation spfactprice3a 

! This equation links domestic and firm demand prices. It holds for Sluggish 

  endowment goods and captures the effect of taxation of firms' usage of 

  primary factors.  (HT#17)! 

    (all,i,ENDW_COMM)(all,j,AGRI_COMM)(all,r,REG) 

    pmesx(i,j,r,"otp") = tfsfp(r) + !tfsfp(i,r)+ !pmest(i,j,r); 

 

(a) Homogenous rate: 

!------------------------------------------------------------------------------! 

! Option 1: Homogenous rate                                                    ! 

!------------------------------------------------------------------------------! 

 

Coefficient (all,r,DS_REG) SH_OTP(r) 

    # Share of domestic support allocated to sfp subsidies #; 

Formula(initial) (all,r,DS_REG) 

    SH_OTP(r) = sum{i,NNATDIS_COMM, sum{j,AGRI_COMM, 

                VFMT(i,j,r)-VFMX(i,j,r,"otp")}} 

              / sum{j, AGRI_COMM, VOM(j,r)}; 

 

Variable(change) (all,r,DS_REG) del_sh_otp(r) 

    # Change in the share of OTP support #; 

 

Update (change) (all,r,DS_REG) 

    SH_OTP(r) = del_sh_otp(r); 

 

Equation o_powerotp (all,r,DS_REG) 

!This equation links the changes in OTP with changes in the corresponding value 

flows! 

      100*(sum{a,AGRI_COMM,VOM(a,r)}*del_sh_otp(r) 

    + SH_OTP(r)*sum{j,AGRI_COMM,VOM(j,r)*[pm(j,r) + qo(j,r)]}) 

    = sum{i,NNATDIS_COMM,sum{j,AGRI_COMM,VFMT(i,j,r)*[pmest(i,j,r) + qfe(i,j,r)] 

    - VFMX(i,j,r,"otp")*[pmesx(i,j,r,"otp") + qfe(i,j,r)]}}; 

 

(b) Other distribution options  

!------------------------------------------------------------------------------! 

! Option 2: other distribution methods                                         ! 

!------------------------------------------------------------------------------! 

 

COEFFICIENT (all,i,NNATDIS_COMM)(all,r,DS_REG) SH_OTP(i,r) 

    # Share of domestic support allocated to sfp subsidies #; 

FORMULA(initial) (all,i,NNATDIS_COMM)(all,r,DS_REG) 

    SH_OTP(i,r) = sum{j,AGRI_COMM, VFMT(i,j,r)-VFMX(i,j,r,"otp")} 

                / sum{j,AGRI_COMM, VOM(j,r)}; 

 

VARIABLE(change) (all,i,NNATDIS_COMM)(all,r,DS_REG) del_sh_otp(i,r) 

    # Percentage change in share of OTP support #; 

 

UPDATE (change) (all,i,NNATDIS_COMM)(all,r,DS_REG) 

   SH_OTP(i,r) = del_sh_otp(i,r); 

 

EQUATION o_powerotp (all,i,NNATDIS_COMM)(all,r,DS_REG) 

    100*(sum{j,AGRI_COMM,VOM(j,r)}*del_sh_otp(i,r) 

    + SH_OTP(i,r)*sum{j,AGRI_COMM,VOM(j,r)*[pm(j,r) + qo(j,r)]}) 

    = sum{j,AGRI_COMM, VFMT(i,j,r)*[pmest(i,j,r) + qfe(i,j,r)] 

    - VFMX(i,j,r,"otp")*[pmesx(i,j,r,"otp") + qfe(i,j,r)]}; 

 

This is an iterative procedure, where the shocks are implemented stepwise, therefore, various 

Gempack command files are used (alterdo.cmf, alterdo1.cmf, …, alterdo8.cmf). Here, only an 

example is shown. Due to the stepwise implementation of shocks some shocks are 

commented out using "!" in the example below. For the SFP please select the options 

accordingly to the distribution method of these payments. 

Example swaps and shocks (alterdo.cmf): 

!------------------------------------------------------------------------------! 

!SFP allocation options:                                                       ! 

!------------------------------------------------------------------------------! 

!------------------------------------------------------------------------------! 

! 1 option: homogenous rate                                                    ! 

!------------------------------------------------------------------------------! 

!swap tfsfp(DS_REG)        = del_sh_OTP(DS_REG); 

!shock del_sh_OTP(DS_REG)  =  select from file ds_shk.har header "sfp"; 

!------------------------------------------------------------------------------! 

! 2. option: other distribution methods                                        ! 

!------------------------------------------------------------------------------! 

!swap tfsfp("land",DS_REG)        = del_sh_OTP("land",DS_REG); 

!swap tfsfp("capital",DS_REG)     = del_sh_OTP("capital",DS_REG); 

!swap tfsfp("unsklab",DS_REG)     = del_sh_OTP("unsklab",DS_REG); 

!swap tfsfp("sklab",DS_REG)       = del_sh_OTP("sklab",DS_REG); 

 

!shock del_sh_OTP("land",DS_REG)   =  select from file ds_shk.har header "sfp"; 

!shock del_sh_OTP("capital",DS_REG)=  select from file ds_shk.har header "sfp"; 

!hock del_sh_OTP("unsklab",DS_REG) =  select from file ds_shk.har header "sfp"; 

!shock del_sh_OTP("sklab",DS_REG)  =  select from file ds_shk.har header "sfp"; 

!------------------------------------------------------------------------------! 

! end option                                                                   ! 

!------------------------------------------------------------------------------! 

!------------------------------------------------------------------------------! 

! SCT                                                                          ! 

!------------------------------------------------------------------------------! 

shock potax_sct(P_AGRI_COMM,DS_REG) = select from file ds_shk.har header "psct"; 

shock potax_sct(PDR_COMM,DS_REG) = select from file ds_shk.har header "psct"; 

shock potax_sct(PFB_COMM,DS_REG) = select from file ds_shk.har header "psct"; 

!------------------------------------------------------------------------------! 

swap tfsct("land",P_AGRI_COMM,DS_REG)    = del_shrland_SCT(P_AGRI_COMM,DS_REG); 

swap tfsct("UnSkLab",P_AGRI_COMM,DS_REG) = del_shrunsk_SCT(P_AGRI_COMM,DS_REG); 

swap tfsct("SkLab",P_AGRI_COMM,DS_REG)   = del_shrskil_SCT(P_AGRI_COMM,DS_REG); 

swap tfsct("capital",P_AGRI_COMM,DS_REG) = del_shrcap_SCT(P_AGRI_COMM,DS_REG); 

swap to(P_AGRI_COMM,DS_REG)              = del_shrout_SCT(P_AGRI_COMM,DS_REG); 

swap tfdmsct(P_AGRI_COMM,DS_REG)         = del_shrint_SCT(P_AGRI_COMM,DS_REG); 

! 

shock del_shrland_SCT(P_AGRI_COMM,DS_REG) 

= select from file ds_shk.har header "shkl"; 

shock del_shrunsk_SCT(P_AGRI_COMM,DS_REG) 

= select from file ds_shk.har header "shku"; 

shock del_shrskil_SCT(P_AGRI_COMM,DS_REG) 

= select from file ds_shk.har header "shks"; 

shock del_shrcap_SCT(P_AGRI_COMM,DS_REG) 

= select from file ds_shk.har header "shkc"; 

shock del_shrout_SCT(P_AGRI_COMM,DS_REG) 

= select from file ds_shk.har header "shko"; 

shock del_shrint_SCT(P_AGRI_COMM,DS_REG) 

= select from file ds_shk.har header "shki"; 

 

swap tfsct("land",PDR_COMM,DS_REG)    = del_shrland_SCT(PDR_COMM,DS_REG); 

swap tfsct("UnSkLab",PDR_COMM,DS_REG) = del_shrunsk_SCT(PDR_COMM,DS_REG); 

swap tfsct("SkLab",PDR_COMM,DS_REG)   = del_shrskil_SCT(PDR_COMM,DS_REG); 

swap tfsct("capital",PDR_COMM,DS_REG) = del_shrcap_SCT(PDR_COMM,DS_REG); 

swap to(PDR_COMM,DS_REG)              = del_shrout_SCT(PDR_COMM,DS_REG); 

swap tfdmsct(PDR_COMM,DS_REG)         = del_shrint_SCT(PDR_COMM,DS_REG); 

 

shock del_shrland_SCT(PDR_COMM,DS_REG) 

= select from file ds_shk.har header "shkl"; 

shock del_shrunsk_SCT(PDR_COMM,DS_REG) 

= select from file ds_shk.har header "shku"; 

shock del_shrskil_SCT(PDR_COMM,DS_REG) 

= select from file ds_shk.har header "shks"; 

shock del_shrcap_SCT(PDR_COMM,DS_REG) 

= select from file ds_shk.har header "shkc"; 

shock del_shrout_SCT(PDR_COMM,DS_REG) 

= select from file ds_shk.har header "shko"; 

shock del_shrint_SCT(PDR_COMM,DS_REG) 

= select from file ds_shk.har header "shki"; 

 

swap tfsct("land",PFB_COMM,DS_REG)    = del_shrland_SCT(PFB_COMM,DS_REG); 

swap tfsct("UnSkLab",PFB_COMM,DS_REG) = del_shrunsk_SCT(PFB_COMM,DS_REG); 

swap tfsct("SkLab",PFB_COMM,DS_REG)   = del_shrskil_SCT(PFB_COMM,DS_REG); 

swap tfsct("capital",PFB_COMM,DS_REG) = del_shrcap_SCT(PFB_COMM,DS_REG); 

swap to(PFB_COMM,DS_REG)              = del_shrout_SCT(PFB_COMM,DS_REG); 

swap tfdmsct(PFB_COMM,DS_REG)         = del_shrint_SCT(PFB_COMM,DS_REG); 

 

shock del_shrland_SCT(PFB_COMM,DS_REG) 

= select from file ds_shk.har header "shkl"; 

shock del_shrunsk_SCT(PFB_COMM,DS_REG) 

= select from file ds_shk.har header "shku"; 

shock del_shrskil_SCT(PFB_COMM,DS_REG) 

= select from file ds_shk.har header "shks"; 

shock del_shrcap_SCT(PFB_COMM,DS_REG) 

= select from file ds_shk.har header "shkc"; 

shock del_shrout_SCT(PFB_COMM,DS_REG) 

= select from file ds_shk.har header "shko"; 

shock del_shrint_SCT(PFB_COMM,DS_REG) 

= select from file ds_shk.har header "shki"; 

!------------------------------------------------------------------------------! 

! ACT + all GCTs                                                               ! 

!------------------------------------------------------------------------------! 

!swap tfsub("land",DS_REG,group)    = del_shrendw("land",DS_REG,group); 

!swap tfsub("UnSkLab",DS_REG,group) = del_shrendw("unsklab",DS_REG,group); 

!swap tfsub("SkLab",DS_REG,group)   = del_shrendw("sklab",DS_REG,group); 

!swap tfsub("capital",DS_REG,group) = del_shrendw("capital",DS_REG,group); 

swap tfdmsub(DS_REG,GROUP)          = del_shrint(DS_REG,GROUP); 

! 

!shock del_shrendw("land",DS_REG,group) 

!=select from file ds_shk.har header "sken"; 

!shock del_shrendw("unsklab",DS_REG,group) 

!=select from file ds_shk.har header "sken"; 

!shock del_shrendw("sklab",DS_REG,group) 

!=select from file ds_shk.har header "sken"; 

!shock del_shrendw("capital",DS_REG,group) 

!=select from file ds_shk.har header "sken"; 

shock del_shrint(DS_REG,GROUP) 

=select from file ds_shk.har header "skin"; 

!------------------------------------------------------------------------------! 

 

After each step of the altertax simulation the shocks have to be re-calculated using ds_shk.tab 

to produce a new shock file.  

6. ds_gtpvew.tab 

This tab-file is used to produce view files of coefficients and tax rates. 

Input: 

 altbase8.upd   Updated GTAP database (GTAPDATA) 

 sets.har   GTAP sets file (GTAPSETS) 

 ds_sets.har   Additional sets for domestic support (DSSETS) 

 default.prm   GTAP parameter file (GTAPPARM) 

 ds_gtpvew.tab   Tab file 

 ds_gtpvew.cmf  Command file 

Output: 

 altbase8T.har   Updated tax rates 

 altbase8V.har   Updated view file 

7. ds_gtap.tab 

This tab-file is used to run the numeraire shock, as a test simulation and implement the new 

closure. 

Input: 

 altbase8.upd   Updated GTAP database (GTAPDATA) 

 sets.har   GTAP sets file (GTAPSETS) 

 ds_sets.har   Additional sets for domestic support (DSSETS) 

 default.prm   GTAP parameter file (GTAPPARM) 

 ds_gtap.tab   Tab file 

 ds_gtap.sti   Stored input file 

Output 

 init.har    Update database 

 init.slc    Header array coefficient file 

 init.sl4    Solution file 

Windows batch file: 

 init.bat 

8. Sensitivity Analysis 

For the sensitivity analysis run ds_shk.tab again to calculate the shock for the removal of the 

single farm payments.  

Input: 

 altbase8.upd   Updated GTAP database (GTAPDATA) 

 ds_gtap.har   Allocation of PSE data (DSDATA) 

 sets.har   GTAP sets file (GTAPSETS) 

 ds_sets.har   Additional sets for domestic support (DSSETS) 

 ds_shk.tab   Tab file 

 ds_shk.cmf   Command file 

Output: 

 ds_shk9.har   Shock file 

Then run ds_gtap.tab to shock the single farm payment to zero. 

Input: 

 altbase8.upd  Updated GTAP database (GTAPDATA) 

 sets.har  GTAP sets file (GTAPSETS) 

 ds_sets.har  Additional sets for domestic support (DSSETS) 

 default.prm  GTAP parameter file (GTAPPARM) 

 ds_shk9.har  Shock file 

 ds_gtap.tab  Tab file 

 ds_gtap.sti  Stored input file 

Output: 

 tfsfpllc.sl4  Solution file 

 tfsfpllc.slc  Header array coefficient file 

 tfsfpllc.har  GTAP updated database 

Windows batch file: 

 p_sfp_0.bat 

You can start the update procedure including the SFP removal using the windows batch 

file p_sfp_0.bat. The results will be written to the "tfsfpllc.har" tfsfpllc.slc and 

"tfsfpllc.sl4". 

 


